

Software-Anleitung

Excel: Starke Konkurrenz für Pivot – wichtige Projekt-Kennzahlen auf einen Blick

Teil 2: Das Cockpit erweitern mit SUMMEWENNS & Jokern

Wie der erste Teil dieses Beitrags zeigt, gibt es außer Pivot-Tabellen auch andere leistungsfähige Techniken, um Projektdaten zu analysieren und das Ergebnis in kompakter Form darzustellen. Die in Teil 1 vorgestellten Techniken basieren auf den Funktionen ZÄHLENWENN und SUMMEWENN, mit denen sich bedingte Abfragen anhand *eines* bestimmten Kriteriums aufbauen lassen.

Nicht immer reicht jedoch ein einziges Kriterium aus, um aussagekräftige Abfragen zu erstellen. Lesen Sie in diesem zweiten Teil, wie Sie Auswertungen mit mehreren Abfragekriterien definieren können. Die dabei verwendeten Funktionen SUMMEWENNS und MITTELWERTWENNS sind ab Excel 2007 standardmäßig verfügbar. Eine weitere hilfreiche Funktion ist SUMMENPRODUKT, mit der Sie auch Abfragen aufbauen können, in denen die Kriterien nicht mit UND, sondern mit ODER verknüpft sind.

	A	B	C	D	E
1	Cockpit zur Projektauswertung für das Fiskaljahr 2011/2012				
2					
3		1	Status A	Status B	Status C
4	Anzahl Projekte nach Status		5	8	5
5					9
6		2	KST	Plankosten	Istkosten
7	Kosten nach Kostenstelle		E5158	17.500 €	16.500 €
8					
9		3	Abteilung	Quartal	Istkosten
10	Istkosten nach Abteilung und Quartal		EDV	4	14.000 €
11					
12		4	Status	von	bis
13	Anzahl Projekte nach Status und Zeitraum		D	01.01.2012	30.04.2012
14					Anz. Projekte
15		5	KST	Quartal	Anz. Projekte
16	Projekte, die in einem Quartal enden		F7*	1	3
17					Istkosten
18		6	Status	Quartal	Anz. Projekte
19	Kosten und Anzahl der Projekte, die in einem		B	3	4
20	Quartal nicht zu Ende gebracht wurden		C		13.480 €
21					
22		7	Status	Abteilung	Ø Istkosten
23	Ø Kosten beendeter Projekte nach Abteilung		D	EDV	7.500 €

Bild 1: Das Beispielcockpit – erstellt mit den Formeln SUMMEWENN, SUMMEWENNS, MITTELWERTWENNS und SUMMENPRODUKT.

Autor



Hildegard Hügemann

Dipl. Informatikerin, seit mehr als 20 Jahren als Software-Entwicklerin und

Trainerin tätig, Schwerpunkte: Microsoft Access und Excel

Kontakt:

hh@huegemann-informatik.de



Dieter Schiecke

Software-Trainer, Consultant und Coach. Spezialgebiet: Visuelle

Umsetzung von Informationen mit Microsoft Office. Chefredakteur der Zeitschrift "PowerPoint aktuell"

Kontakt: dieter@schiecke.biz

Mehr Informationen unter:

projektmagazin.de/autoren

ähnliche Artikel

in den Rubriken:

[Microsoft Excel](#)

[Kostencontrolling](#)

Dieser Beitrag knüpft unmittelbar an den ersten Teil an und stellt die Techniken für die Abfragen 3 bis 7 des dort beschriebenen Beispiels vor (Bild 1). Die Tabelle mit den Projektdaten des Beispiels (Bild 2) können Sie zusammen mit diesem Beitrag herunterladen.

Projekt	Abteilung	KST	Status	Beginn	Ende	Plankoste	Istkosten	Quartal
Einführung Business Contact Manager	FE	F6220	D	10.10.2011	25.11.2011	8.800 €	9.000 €	4
Update BizTalk Server	EDV	E5160	A	11.10.2011	31.10.2011	9.800 €	11.000 €	4
Dynamics AX 2012	EDV	E5159	B	18.10.2011	20.12.2011	5.600 €	3.000 €	4
Pilotprojekt Granulat HF422	FE	F7228	D	28.10.2011	13.01.2012	3.000 €	3.500 €	1

Bild 2: Auszug aus der Datentabelle des Beispiels.

Abfrage 3: Wie hoch sind die Istkosten in einem Quartal für eine bestimmte Abteilung?

Eine typische Fragestellung in unserem Beispiel könnte lauten: Wie viel Euro wurden im 4. Quartal für EDV-Projekte ausgegeben? In diesem Fall müssen also mehrere Kriterien geprüft werden, bevor summiert wird: Die Abteilung und der Zeitraum (hier das Quartal).

Mit Hilfe der Funktion SUMMEWENNS können Sie diese Frage beantworten. So geht's:

1. Wählen Sie zunächst im Auswertungsbereich in Zelle B10 in der Dropdown-Liste den Eintrag "EDV". Geben Sie rechts daneben in C10 die Zahl für das gesuchte Quartal ein – hier also eine "4".
2. Markieren Sie D10. Hier soll die Summe der Kosten berechnet werden, für die die Kriterien in B10 und C10 zutreffen.
3. Tippen Sie "=SU", wählen Sie aus der angezeigten Funktionsliste den Eintrag SUMMEWENNS aus und übernehmen Sie die Funktion per Tab-Taste in die Bearbeitungsleiste.
4. Zur Eingabe des ersten Arguments ("*Summe_Bereich*") positionieren Sie in der Datentabelle den Mauszeiger über der Spalte "Istkosten" und klicken Sie, sobald dieser sich in einen kleinen schwarzen Pfeil verwandelt. Die angezeigte Formel lautet jetzt =SUMMEWENNS(*Projekte[Istkosten]*).
5. Setzen Sie ein Semikolon und markieren Sie auf die gleiche Weise die Spalte "Abteilung", um das zweite Argument festzulegen ("*Kriterien_Bereich1*"). In der Bearbeitungsleiste steht jetzt als vorläufige Formel: =SUMMEWENNS(*Projekte[Istkosten]*; *Projekte[Abteilung]*).
6. Nach einem weiteren Semikolon klicken Sie auf Zelle B10. Dort steht das erste Kriterium – hier "EDV" für die zu berücksichtigende Abteilung. Die Formel lautet nun: =SUMMEWENNS(*Projekte[Istkosten]*; *Projekte[Abteilung]*; B10).
7. Nach einem erneuten Semikolon lassen Sie das Argument für *Kriterien_Bereich2* folgen. Markieren Sie dazu wieder mit dem kleinen schwarzen Pfeil die Spalte "Quartal". Die Formel lautet jetzt: =SUMMEWENNS(*Projekte[Istkosten]*; *Projekte[Abteilung]*; B10; *Projekte[Quartal]*).

8. Fügen Sie nach einem weiteren Semikolon das Argument *Kriterien2* hinzu, indem Sie auf C10 klicken, also die Zelle mit der Ziffer des gewünschten Quartals. Schließen Sie mit Enter ab. Die komplette Formel ist:

`=SUMMEWENNS(Projekte[Istkosten];Projekte[Abteilung];B10;Projekte[Quartal];C10)`

f	<code>=SUMMEWENNS(Projekte[Istkosten];Projekte[Abteilung];B10;Projekte[Quartal];C10)</code>				
	B	C	D	E	F
	3	Abteilung	Quartal	Istkosten	
	SUMMEWENNS(Summe_Bereich; Kriterien_Bereich1; Kriterien1; [Kriterien_Bereich2; Kriterien2]; [Kriterien_Bereich3; Kriterien3]; ...)				

Bild 3: Mit dieser SUMMEWENNS-Funktion werden zwei Auswahlkriterien beim Addieren von Werten berücksichtigt.

Abfrage 4: Wie viele Projekte wurden in den ersten vier Monaten von 2012 abgeschlossen?

Umfasst der abzufragende Zeitraum ein komplettes Quartal, kann in der Formel auf die Informationen in der Spalte "Quartal" zurückgegriffen werden, um den Zeitraum anzugeben. Doch was, wenn der zu betrachtende Zeitraum nicht genau einem Quartal entspricht, sondern kürzer oder länger ist? So kann beispielsweise im Mai die Aufgabe anstehen, die Anzahl der Projekte zu ermitteln, die seit Jahresbeginn bis zum Ende des Monats April abgeschlossen wurden.

Hier muss die Formel drei Bedingungen berücksichtigen: Den Projektstatus (in dem Fall D für "Fertig"), den Beginn sowie das Ende des gewünschten Zeitraums. Bild 4 zeigt die Ausgangssituation. Zur Lösung der Aufgabe verwenden Sie diesmal die Funktion ZÄHLENWENNS.

1. Geben Sie zunächst in den Zellen B13, C13 und D13 die Kriterien ein (Bild 4).

2. Markieren Sie Zelle E13. Hier soll die Anzahl der Projekte berechnet werden, die alle drei Kriterien erfüllen.

3. Tippen Sie "=Z" ein. Wählen Sie mit der Pfeiltaste nach unten den Eintrag ZÄHLENWENNS und holen Sie diesen per Tab-Taste in die Bearbeitungsleiste.

4. Klicken Sie zum Festlegen des ersten Arguments (*Kriterienbereich1*) über der Spalte "Status" mit dem kleinen schwarzen Pfeil. Setzen Sie ein Semikolon und klicken Sie anschließend auf B13, in der das Kriterium steht. Die angezeigte Formel lautet: `=ZÄHLENWENNS(Projekte[Status];B13)`.

5. Tippen Sie ein weiteres Semikolon ein. Für das Argument *Kriterienbereich2* markieren Sie mit dem kleinen schwarzen Pfeil die Spalte "Beginn".

6. Nach einem Semikolon geben Sie für das Argument *Kriterien2* die Zeichenfolge `">="&` ein und klicken auf Zelle C13. Die Formel lautet nun: `=ZÄHLENWENNS(Projekte[Status];B13;Projekte[Beginn];">="&C13)`.

	B	C	D	E
11				
12	Status	von	bis	Anz. Projekte
13	D	01.01.2012	30.04.2012	

Bild 4: Beim Berechnen der Anzahl werden drei Kriterien berücksichtigt.

7. Ergänzen Sie nach einem weiteren Semikolon die Bezüge für die Argumente *Kriterienbereich3* und *Kriterien3*. Markieren Sie dafür mit dem schwarzen Pfeil die Spalte "Ende". Tippen Sie nach einem Semikolon die Zeichenfolge "<="& ein. Klicken Sie dann auf Zelle D13. Schließen Sie mit "Enter" ab. Die fertige Formel lautet:

=ZÄHLENWENNS(Projekte[Status];B13;Projekte[Beginn];">="&C13;Projekte[Ende];"<="&D13)					
	B	C	D	E	F
	Status	von	bis	Anz. Projekte	
	D	01.01.2012	30.04.2012	3	

Bild 5: Die fertige Formel ZÄHLENWENNS berücksichtigt drei Bedingungen.

=ZÄHLENWENNS(Projekte[Status];B13;Projekte[Beginn];">="&C13;Projekte[Ende];"<="&D13).

Kriterien flexibel definieren mit Hilfe von Vergleichsoperatoren

Neben absoluten Werten (z.B. "1") oder Zellbezügen (z.B. "C13") können Sie bei der Angabe eines Kriteriums auch Vergleichsoperatoren verwenden. Tabelle 1 gibt einen Überblick über die Vergleichsoperatoren, Tabelle 2 liefert Beispiele für deren Verwendung.

Abfrage 5: Wie viele Projekte eines bestimmten Kostenstellenbereichs enden in einem Quartal und wie hoch sind die dabei angefallenen Kosten?

Sie können diese Abfrage im Prinzip wieder wie zuvor mit den Funktionen ZÄHLENWENNS und SUMMEWENNS aufbauen. Neu ist diesmal jedoch

Vergleichsoperator	Bedeutung
=	Gleich
>	Größer
<	Kleiner
>=	Größer gleich
<=	Kleiner gleich
<>	Ungleich

Tabelle 1: Vergleichsoperatoren.

Kriterium	Gezählt werden alle Zellen ...
=ZÄHLENWENN(Projekte[Abteilung];"EDV")	im Bereich Abteilung, die den Text "EDV" enthalten
=ZÄHLENWENN(Projekte[Abteilung];"<>EDV")	im Bereich Abteilung, in denen nicht "EDV" steht
=ZÄHLENWENN(Projekte[Istkosten];">5000")	im Bereich Istkosten, die den Wert 5000 übersteigen
=ZÄHLENWENN(Projekte[Quartal];"=4")	im Bereich Quartal, die den Eintrag 4 besitzen (alle Projekte, die im 4. Quartal beendet wurden)
=ZÄHLENWENN(Projekte[Quartal];"<3")	im Bereich Quartal, die einen Eintrag kleiner als 3 besitzen (also alle Projekte, die im 1. oder im 2. Quartal beendet wurden)
=ZÄHLENWENN(Projekte[Projekt];"*Office*")	im Bereich Projekt, in denen der Text "Office" vorkommt
=ZÄHLENWENN(Projekte[KST];"=?7*")	im Bereich KST, die an der zweiten Stelle eine 7 aufweisen
=ZÄHLENWENN(Projekte[KST];"<>"& B16)	im Bereich KST, die nicht der Kostenstelle in der Zelle B16 entsprechen
=ZÄHLENWENN(Projekte[KST];"<>"& "")	im Bereich KST, die nicht belegt sind

Tabelle 2: Beispiele für den Einsatz von Vergleichsoperatoren (hier am Beispiel der Funktion ZÄHLENWENN).

die Aufgabe, Kostenstellen zu berücksichtigen, die in einem bestimmten Bereich liegen – beispielsweise alle Kostenstellen für die Abteilung FE, die mit der Ziffer "7" beginnen.

Das gelingt Ihnen, indem Sie mit sog. "Joker"-Zeichen arbeiten.

Mit Joker-Zeichen die Eingabe von Kriterien für die Auswertung variabel halten

Es gibt zwei verschiedene Joker-Zeichen in Excel: das Fragezeichen (?) und den Stern (*). Das "?" nutzen Sie, um *ein* beliebiges Zeichen zu ersetzen, das Zeichen "*" verwenden Sie, wenn Sie beliebige und beliebig viele Zeichen ersetzen möchten.

Hier einige Beispiele für den Einsatz der Joker und für deren unterschiedliche Wirkung:

- *optimierung* im Bereich Projekt liefert alle Projekte, bei denen es um Optimierung geht, die also irgendwo im Projektnamen den Text "Optimierung" besitzen.
- F* im Bereich KST liefert alle Kostenstellen aus dem Bereich Forschung & Entwicklung, also alle die mit F beginnen.
- ?7* im Bereich KST liefert alle Kostenstellen, die mit einem beliebigen Zeichen beginnen, an zweiter Stelle eine 7 und dann wieder beliebige Zeichen aufweisen.

Ausgerüstet mit diesem Wissen können Sie die eingangs formulierte Abfrage lösen. Gehen Sie wie folgt vor, um zu ermitteln, wie viele Projekte eines bestimmten Kostenstellenbereichs in einem bestimmten Quartal enden und welche Kosten dabei anfallen:

1. Geben Sie zunächst in Zelle B16 die Zeichenfolge "F7*" ein. Alle Kostenstellen, die mit "F7" beginnen, werden somit berücksichtigt. Tragen Sie in C16 die Ziffer für das gesuchte Quartal ein – hier "1".

=SUMMEWENNS(Projekte[Istkosten];Projekte[KST];B16;Projekte[Quartal];C16)						
	B	C	D	E	F	G
	KST	Quartal	Anz. Projekte	Istkosten		
	F7*	1	3	7.000 €		

Bild 7: Die fertige Abfrage mit der Funktion SUMMEWENNS.

2. Die Formel, die Sie in D16 eintragen, lautet: =ZÄHLENWENNS(Projekte[KST];B16;Projekte[Quartal];C16).
3. In E16 verwenden Sie =SUMMEWENNS(Projekte[Istkosten];Projekte[KST];B16;Projekte[Quartal];C16).

	B	C	D	E
14				
15	KST	Quartal	Anz. Projekte	Istkosten
16	F7*	1		

Bild 6: Beim Definieren der Suchkriterien können auch Joker verwendet werden.

Abfrage 6: Wie hoch sind Kosten und Anzahl der Projekte, die in einem bestimmten Quartal nicht zu Ende gebracht wurden?

Nicht immer läuft alles wie geplant. Daher gibt es auch Projekte, die nicht zu Ende geführt werden – entweder weil sie eingefroren (Status B) oder weil sie eingestellt wurden (Status C). Über den Status lässt sich ermitteln, welche Projekte das sind: Es sind die mit dem Status B oder dem Status C. In der Abfrage müssen Sie die Bedingungen also in diesem Fall mit einem ODER verknüpfen und nicht wie bisher mit einem UND.

Die Funktionen SUMMEWENNS und ZÄHLENWENNS sind für eine solche Abfrage nicht geeignet, da sie die Kriterien automatisch mit einem UND verknüpfen. Deshalb soll bei dieser Abfrage eine andere Funktion zum Einsatz kommen: SUMMENPRODUKT. So geht's:

1. Legen Sie zunächst in den Zeilen 19 und 20 der Spalten für "Status" und für "Quartal" die Kriterien fest.

2. Markieren Sie D19. Geben Sie dort folgende Formel ein:

=SUMMENPRODUKT(((Projekte[Status]=B19)+(Projekte[Status]=B20))*(Projekte[Quartal]=C19)).

	B	C	D	E
17				
18	Status	Quartal	Anz. Projekte	Istkosten
19	B	1		
20	C			

Bild 8: In D20 und E20 sollen Auswertungen erfolgen, die die Kriterien in B19, B20 und C19 berücksichtigen.

=SUMMENPRODUKT(((Projekte[Status]=B19)+(Projekte[Status]=B20))*(Projekte[Quartal]=C19)*Projekte[Istkosten])						
	B	C	D	E	F	G
	Status	Quartal	Anz. Projekte	Istkosten		
	B	3	4	13.480 €		
	C					

Bild 9: Die Funktion SUMMENPRODUKT macht eine ODER-Verknüpfung von Kriterien ohne zusätzliche Hilfsspalten möglich.

3. Wechseln Sie rechts daneben in Zelle E19. Tragen Sie folgende Formel ein:

=SUMMENPRODUKT(((Projekte[Status]=B19)+(Projekte[Status]=B20))*(Projekte[Quartal]=C19)*
Projekte[Istkosten]).



Mit der Funktion SUMMENPRODUKT ist es sowohl möglich eine Anzahl als auch eine Summe zu berechnen.

Zur Erläuterung:

- In der Funktion SUMMENPRODUKT lässt sich mit Wahrheitswerten (WAHR = 1, FALSCH = 0) rechnen. Das wird genutzt, um eine Summe zu bilden oder eine Anzahl zu berechnen.
- Beispielsweise liefert der Ausdruck (Projekte[Status]=B19)+(Projekte[Status]=B20) in jeder Zeile entweder den Wert 1 (wenn eine der beiden Bedingungen WAHR ist und die andere FALSCH) oder den Wert 0 (wenn

beide Bedingungen FALSCH sind). Beide Bedingungen gleichzeitig können nicht wahr sein, da B19 und B20 verschiedene Werte enthalten.

- Der Ausdruck `Projekte[Quartal]=C19` liefert WAHR und damit den Wert 1, wenn in der entsprechenden Zeile das gesuchte Quartal steht.
- Werden diese beiden Ausdrücke miteinander multipliziert, ergibt sich als Ergebnis nur dann eine 1, wenn beide Faktoren dem Wert 1 entsprechen ($1 \cdot 1 = 1$), also der Status B ODER C ist UND das Quartal mit dem in C19 übereinstimmt. SUMMENPRODUKT addiert dann die Ergebnisse pro Zeile, in diesem Fall die Einsen. Dies liefert die Anzahl der zutreffenden Projekte.
- Wird das Ergebnis dann noch mit dem Ausdruck `Projekte[Istkosten]` multipliziert, werden für alle "passenden" Projekte nicht die Einsen addiert, sondern die jeweiligen Istkosten.

Abfrage 7: Wie hoch sind durchschnittlich die Kosten beendeter Projekte pro Abteilung?

Führt eine Abteilung im Jahr mehrere Projekte durch und möchte sie die Kosten für die bereits beendeten Projekte ermitteln, lässt sich das mit einer entsprechenden Abfrage und den bisher vorgestellten Techniken leicht umsetzen. Doch was, wenn in der Auswertung zu Vergleichszwecken auch die Information erscheinen soll, wie hoch die durchschnittlichen Projektkosten in dieser Abteilung sind?

Sollen Durchschnittswerte mit mehreren Einschränkungen ermittelt werden – hier nur für fertig gestellte Projekte (also Status D) und nur für eine bestimmte Abteilung – kommt die Funktion MITTELWERTWENNS ins Spiel. Sie können eine entsprechende Abfrage wie folgt anlegen:

1. Geben Sie in den Zellen B23 und C23 die gewünschten Kriterien ein: z.B. Status "D" und Abteilung "EDV".

2. Markieren Sie Zelle D23. Hier soll der Durchschnittswert der Kosten unter Berücksichtigung der eben definierten Bedingungen in B23 und C23 ermittelt werden.

=MITTELWERTWENNS(Projekte[Istkosten];Projekte[Status];B23;Projekte[Abteilung];C23)				
	B	C	D	E
	Status	Abteilung	Ø Istkosten	
	D	EDV	7.500 €	

Bild 10: MITTELWERTWENNS ermittelt, wie hoch in der Abteilung EDV durchschnittlich die Kosten für fertig gestellte Projekte sind.

3. Stellen Sie dann ähnlich wie bei der Funktion SUMMEWENNS die Formel für MITTELWERTWENNS auf. Sie lautet:
`=MITTELWERTWENNS(Projekte[Istkosten];Projekte[Status];B23;Projekte[Abteilung];C23)`.

Bild 10 zeigt die fertige Lösung für die Abteilung EDV. Die Formel ist wieder gelb markiert.

Fazit

Mit ZÄHLENWENN und SUMMEWENN ermitteln Sie die Anzahl oder die Summe von Daten, die *ein* bestimmtes Kriterium erfüllen. Die Funktionen ZÄHLENWENNS und SUMMEWENNS setzen Sie ein, wenn beim Zählen und Summieren von Daten *ein oder mehr als ein* Kriterium zu berücksichtigen ist.

MITTELWERTWENNS verwenden Sie, wenn es darum geht, Durchschnittswerte für Daten zu berechnen, auf die bestimmte Kriterien (ein oder mehrere) zutreffen.

SUMMENPRODUKT ist eine flexible Matrixfunktion, mit der Sie – im Unterschied zu den bisher genannten Funktionen – nicht nur UND-, sondern auch ODER-Bedingungen abfragen und Bedingungen über berechnete Ausdrücke zusammenstellen können. Und dies ohne das Anlegen zusätzlicher Hilfsspalten. Sie eignet sich sowohl für die Anzahl- wie auch für die Summenbildung von Konstanten und berechneten Werten.

Ausblick: Wie leistungsfähig sind Datenbank-Funktionen und Spezialfilter?

Wussten Sie, dass Sie beim Einsatz von DB-Funktionen mit kürzeren Formeln zum gleichen Ergebnis gelangen? Beispielsweise führt die in Bild 11 gelb markierte Formel zum gleichen Ergebnis wie die lange Formel

$$=SUMMENPRODUKT(((Projekte[Status]=B19)+(Projekte[Status]=B20))*(Projekte[Quartal]=C19)*Projekte[Istkosten])$$

Lesen Sie im nächsten Beitrag, wie Sie mit den Funktionen DBSUMME, DBANZAHL und DBMITTELWERT ebenso informative Auswertungen erstellen wie mit SUMMEWENNS & Co. Erfahren Sie dabei auch, wie Sie flexible Kriterienbereiche aufbauen, die dann für den Einsatz des Spezialfilters zur Verfügung stehen.

E19 $=DBSUMME(Projekte;"Istkosten";B18:C20)$

	A	B	C	D	E
Cockpit zur Projektauswertung für das Fiskaljahr 2011/2012					
1		1 Status	Status	Status	Status
2		A	B	C	D
3		5	8	5	9
4	Anzahl Projekte nach Status				
5					
6		2 KST	Plankosten	Istkosten	
7		E5158	17.500 €	16.500 €	
8	Kosten nach Kostenstelle				
9		3 Abteilung	Quartal	Istkosten	
10		EDV	4	14.000 €	
11	Istkosten nach Abteilung und Quartal				
12		4 Status	Beginn	Ende	Anz. Projekte
13		D	>=01.01.2012	<=30.04.2012	3
14	Anzahl Projekte nach Status und Zeitraum				
15		5 KST	Quartal	Anz. Projekte	Istkosten
16		F7*	1	3	7.000 €
17	Projekte, die in einem Quartal enden				
18		6 Status	Quartal	Anz. Projekte	Istkosten
19		B	3	4	13.480 €
20	Kosten und Anzahl der Projekte, die in einem Quartal nicht zu Ende gebracht wurden				
21		C	3		
22		7 Status	Abteilung	Ø Istkosten	
23	Ø Kosten beendeter Projekte nach Abteilung	D	EDV	7.500 €	

Bild 11: Dieses Auswertungs-Cockpit ist komplett mit den wenig bekannten Datenbank-Funktionen von Excel ausgestattet.

Erfahren Sie, wie Sie mit dem Spezialfilter Ihre Auswertungen noch informativer machen, indem Sie die einzelnen Ergebnisse bei Bedarf auch im Detail belegen.

Hat Ihnen dieser Artikel gefallen?

Bewerten Sie ihn im Projekt Magazin online und teilen Sie so Ihre Meinung anderen Lesern mit. Wählen Sie dazu den Artikel im Internet unter www.projektmagazin.de/ausgaben/2012 oder klicken Sie [hier](#), um direkt zum Artikel zu gelangen.